NASA TM-84488

NASA Technical Memorandum 84488

PRELIMINARY SIZING OF VIBRATION ABSORBER FOR SPACE MAST STRUCTURES

Michael F. Card, Harvey G, McComb, Jr., and Susan W. Peebles

May 1982

National Aeronautics and Space Administration

Langley Research Center
Hampton, Virginia 23665

PRELIMINARY SIZING OF VIBRATION ABSORBER FOR SPACE MAST STRUCTURES

By

Michael F. Card, Harvey G. McComb, Jr., and Susan W. Peebles

SUMMARY

A simple method of sizing a vibration absorber for a large, cantilevered flexible mast is presented. The method is based on Den Hartog's vibration absorber theory for two-degree-of-freedom systems. Generalized design curves are presented as well as specific numerical results for a candidate space experiment in which a long flexible antenna mast is attached to the shuttle orbiter and dynamically excited by orbiter accelerations. Results indicate that for large flexible masts, the mass of the vibration absorber required to meet stringent tip deflection tolerances becomes prohibitively large.

N82-28346 #

INTRODUCTION

The technical feasibility of deploying and operating large antennas in space is a current thrust of NASA research. A typical system under study is the 55-meter "offset wrap rib" concept shown in figure 1 (see ref. 1). Because of the large aperture and feed mast dimensions, there is a need to understand how to control dynamic response of flexible bodies within stringent deflectional tolerances.

In the present paper, a simplified method is developed to size passive vibration absorbers for large flexible space antenna masts. The analysis is based on vibration absorber theory developed by Den Hartog in reference 2. Approximate equations are developed to determine the mass, stiffness and damping ratios required for absorbers to meet specified dynamic deflection requirements. Results are presented for selected designs of large antenna feed masts deployed as cantilevered beams from the space shuttle orbiter.

SYMBOLS

c_A	damping in absorber
c_{cr}	critical damping for absorber, $2M_A\omega_A$
E	modulus of elasticity
I	area moment of inertia of beam or mast cross-section
I_M	mass moment of inertia
K_A	spring constant for absorber

K_B	equivalent spring constant for beam tip lateral deflection
ℓ	length of beam or mast
M	mass of antenna feed
M_A	mass of absorber
m	mass of beam or mast without feed
P	statically equivalent lateral tip load, amplitude of dynamic force
w	tip lateral deflection amplitude due to dynamic force
w_{st}	static tip lateral deflection
μ	ratio of absorber mass to beam plus feed mass, $\dfrac{M_A}{M + \frac{1}{3} m}$
$\ddot{\Theta}$	amplitude of rotational acceleration at root of beam or mast
ω	circular frequency of forcing function
ω_A	circular frequency of absorber
ω_B	circular frequency of beam plus feed mass

SIMPLIFIED MODEL AND ANALYSIS

Analysis Assumptions

The analysis is based on the vibration absorber theory given in reference 2. For a beam of length ℓ attached to the shuttle orbiter, shown idealized in figure 2, the equivalent spring coefficient K_B for the beam lateral tip deflection is given (ref. 2) as

$$K_B = \frac{3EI}{\ell^3} \tag{1}$$

where EI is the beam bending stiffness.

The forces acting on the beam can be converted into an equivalent static lateral force P at the beam tip by taking moments about the beam root. Thus,

$$P = \frac{I_M \ddot{\Theta}}{\ell} \tag{2}$$

If the mass moment of inertia of the beam plus tip or feed mass M about the root is approximated as

$$I_M = \frac{1}{3} m \ell^2 + M\ell^2 \tag{3}$$

then the force P is

$$P = (\frac{1}{3} m + M) \ell \ddot{\Theta} \tag{4}$$

A vibration absorber with mass M_A, stiffness K_A, and damping constant c_A can be attached to the simplified beam plus feed mass as indicated in figure 2(c). The beam and feed mass system and attached vibration absorber is represented by the two-degree-of-freedom system in figure 2(c).

Governing Equations

It is shown in reference 2 that with two-degree-of-freedom systems on a plot of amplitude as a function of frequency (see fig. 3), there exist two "fixed points." The locations of the points P and Q in figure 3 are independent of damping, i.e., for any value of damping the curves pass through P and Q (see also ref. 3). This fact is used in reference 2 to develop a theory for tuning an absorber in such a way as to minimize the vibration amplitude of the main system. Also shown in reference 2 is the theory for the case in which the absorber frequency is simply matched to that of the main system. In the present paper, equations are presented for both of these cases. The case for which the absorber is tuned for minimum amplitude of the main system is denoted "best-tuned." The case for which the absorber is matched to the main system is denoted "frequency-matched."

The deflection requirement at the tip for the beam under dynamic excitation $P \sin \omega t$ is specified as w. Then the theory of reference 2 yields for the best-tuned case:

$$\frac{w}{w_{st}} = \sqrt{1 + \frac{2}{\mu}} \tag{5}$$

and for the frequency-matched case:

$$\frac{w}{w_{st}} = \frac{1}{-\mu + (1 + \mu)\sqrt{\frac{\mu}{2 + \mu}}} \tag{6}$$

where

$$w_{st} = \frac{P}{K_B} \tag{7}$$

and

$$\mu = \frac{M_A}{M + \frac{1}{3}m} \tag{8}$$

Best-tuned absorber.- For the best-tuned absorber case, the mass ratio of the absorber to beam plus feed mass can be expressed

$$\mu = \frac{2}{\left(\frac{w}{w_{st}}\right)^2 - 1} \tag{9}$$

The natural circular frequency of the absorber should be

$$\omega_A = \frac{1}{1 + \mu} \tag{10}$$

and the damping ratio can be found from

$$\left(\frac{c_A}{c_{cr}}\right)^2 = \frac{3\mu}{8\ (1 + \mu)^3} \tag{11}$$

<u>Frequency-matched absorber</u>.- If the natural circular frequency of the absorber is matched to that of the beam plus feed mass system, then the mass ratio is the smallest root of

$$\mu = \frac{\left[\left(\frac{w}{w_{st}}\right)^2 - 4\frac{w}{w_{st}} - 1\right] \pm \sqrt{\left[\left(\frac{w}{w_{st}}\right)^2 - 4\frac{w}{w_{st}} - 1\right]^2 - 16\frac{w}{w_{st}}}}{4\frac{w}{w_{st}}} \tag{12}$$

The natural circular frequency of the absorber is

$$\omega_A = \omega = \sqrt{\frac{K_B}{M + \frac{1}{3}m}} \tag{13}$$

The damping ratio can be found from

$$\left(\frac{c_A}{c_{cr}}\right)^2 = \frac{\mu\,(3 + \mu)\left(1 + \sqrt{\frac{\mu}{2 + \mu}}\right)}{8\,(1 + \mu)} \tag{14}$$

A typographical error in equation (3.37) in reference 2 (bottom of p. 103) has been corrected by using reference 4.

CALCULATED RESULTS

General Design Curves

A set of general design curves developed from equations (5) and (6) is presented in figure 4. The curves have been extended beyond the practical ranges shown in reference 2 to indicate overall trends.

For the frequency-matched absorber case, there is no absorber design existing if the ratio of specified tip deflection to static tip deflection is less than 5.8 (i.e., $\frac{w}{w_{st}} \leq 5.8$, see fig. 4). Using the definition of w_{st} for a given beam and loading, there is a critical length given by

$$\ell_{cr} \geq \sqrt[4]{\frac{0.52\ w\ EI}{(M + \frac{1}{3}m)\ \ddot{\Theta}}} \tag{15}$$

beyond which no design exists.

For the best-tuned absorber, the design curve indicates that the mass of the absorber becomes prohibitively large when $\frac{w}{w_{st}} \rightarrow 1$. Thus for this case there is a critical length given by

$$\ell_{cr} \rightarrow \sqrt[4]{\frac{3w\ EI}{(M + \frac{1}{3}m)\ \ddot{\Theta}}} \tag{16}$$

at which the absorber mass approaches infinity.

Space Mast Application

A study was conducted of the absorber properties required for a large antenna feed mast described in figure 5. The mast proportions are based on a representative graphite/epoxy mast being investigated as a space flight experiment in reference 5. Typical feed mast requirements being considered are control of the dynamic response of the tip to within $\pm$ 10 cm ($w = 0.10$ m) under simulated orbiter disturbances of $\ddot{\Theta} = 0.001$ radians/sec^2. In the study, the mass of the absorber to be placed at the tip of the mast was calculated as a function of the length of the mast or beam.

Typical results of the study for best-tuned absorbers and for frequency-matched absorbers are presented in figures 6 and 7. In the best-tuned case, the mass of the absorber becomes prohibitively large as the mast length approaches 62 m. To confirm the validity of the fixed-point theorem in determining best-tuned absorber proportions, the behavior of the vibration absorber was investigated for large absorber mass. Typical response results for the best-tuned absorber are shown in figure 8. These results show that even for large absorber mass the two peak amplitude responses are well matched, as assumed in the fixed-point theorem. In the frequency-matched case, no real solution of equation (12) exists beyond $\ell = 40.5$ m, a result consistent with figure 4 and equation (15).

Results shown in figures 6 and 7 indicate that when the required dynamic deflection approaches the static deflection of the mast, the mast flexibility leads to prohibitively large control forces. A brief exploration of the idea of changing the position of the absorber was undertaken. Based on static force considerations, it appears that moving the absorber inboard from the mast tip does not lead to improved performance. The increase in force required by an inboard absorber to produce a constant tip deflection is shown in figure 9. The calculation is based on cantilever beam statics and shows that appreciable force increases (and hence mass increases) would be required for a single inboard absorber.

CONCLUDING REMARKS

A simplified method has been developed to estimate the mass of a vibration absorber necessary to control the dynamic response of a cantilevered beam tip to a specified dynamic deflection. The method is based on two-degree-of-freedom analyses developed by Den Hartog (ref. 2) for vibration absorbers.

Numerical studies have been conducted on a representative space antenna mast flying attached to the space shuttle orbiter and excited by orbiter disturbance forces. Results of the study show that when the static deflection of the mast is approximately that of the required dynamic deflection, the absorber mass becomes prohibitively large.

Because of the approximate nature of the results presented in this study, more refined studies of multi-degree of freedom models using more accurate beam theory is needed. Furthermore, similar studies should be performed on the effectiveness of other types of vibration controllers, such as momentum wheels or paired gyros (ref. 6). The authors believe more research is needed to develop a physical perspective on practical sizes for vibration controllers in large space antenna applications.

REFERENCES

1. Golden, C. T.; Lackey, J. A.; and Spier, E. E.: Configuration Development of the Land Mobile Satellite System (LMSS) Spacecraft. Large Space Systems Technology, NASA CP-2215, Part 2, November 1981, pp. 711-761.

2. Den Hartog, J. P.: Mechanical Vibrations. McGraw-Hill, Fourth Edition, New York, NY, 1956.

3. Crossley, F. R. E.: Systems of Several Degrees of Freedom. Handbook of Engineering Mechanics by W. Flugge, Chapter 57, pp. 25-26, McGraw-Hill, New York, NY, 1962.

4. Brock, J. E.: A Note on the Damped Vibration Absorber. Trans. ASME, 1946, p. A-284.

5. Jenkins, L.: Large Space Structures Shuttle Flight Experiment. Large Space Systems Technology, NASA CP-2215, Part 2, November 1981, pp. 893-909.

6. Aubrun, J. N.; and Margulies, G.: Gyrodampers for Large Space Structures. NASA CR-159171, February 1979.

APPENDIX

Computer Programs for Absorber Designs

The following programs were written in FORTRAN IV and run on the CDC CYBER using the Network Operating System (NOS). The plots were generated by the Tektronix PLOT10 graphics package with a NASA-Langley interface.

The two programs read NAMELIST input from TAPE5 and write output values to TAPE6. Values of MU and L are written to TAPE3 to be read back and plotted.

SYMBOL IN TEXT	INPUT		
EI	EI	BENDING STIFFNESS OF BEAM	N-M**2
ℓ	L	BEAM LENGTH	M
m	MB	MASS OF BEAM	KG
M	MF	MASS OF FEED	KG
w	W	BEAM TIP EXCURSION	M
$\ddot{\Theta}$	ACC	BASE ROTATIONAL ACCELERATION	RAD/SEC**2
	NP	NUMBER OF FREQUENCY RATIO POINTS	INTEGER
	DELTA	INCREMENT OF FR/FR1	REAL
	FINIT	INITIAL VALUE OF FR	REAL
	OUTPUT		
$M + \frac{1}{3} m$	M1	EQUIVALENT TIP MASS	KG
K_B	K1	EQUIVALENT BEAM SPRING	N/M
w_{st}	WST	STATIC DEFLECTION	M
μ	MU	M2/M1	KG
M_A	M2	ABSORBER MASS	KG
c_A	C2	ABSORBER DAMPING RATIO	
ω	FR	FREQUENCY OF FORCING FUNCTION	RAD/SEC
ω_B	FR1	CIRCULAR FREQUENCY OF BEAM PLUS TIP MASS	RAD/SEC
ω_A	FR2	CIRCULAR FREQUENCY OF ABSORBER	RAD/SEC
	F1	FREQUENCY OF BEAM PLUS TIP MASS	HERTZ
	F2	ABSORBER FREQUENCY	HERTZ
	FR/FR1, W/WST	ARRAYS TO BE PLOTTED	
	FMAX, WMAX	MAXIMUM VALUES OF FR/FR1 AND W/WST	

EQUATIONS FOR BEST TUNED AND FREQUENCY MATCHED ABSORBERS

$$M1 = MF + \frac{1}{3} MB$$

$$K1 = \frac{3\ EI}{L^3}$$

$$WST = \frac{M1 * L * ACC}{K1}$$

BEST TUNED

$$MU = \frac{2}{(W/WST)^2 - 1}$$

$$FB = \frac{1}{1 + MU}$$

$$C1 = \frac{3 * MU}{8 * (1 + MU)^3}$$

FREQUENCY MATCHED

$$X = \frac{W}{WST}$$

$$MU = \frac{X^2 - 4X - 1 \pm \sqrt{(X^2 - 4X - 1)^2 - 16X}}{4X}$$

Select smallest positive root.
If MU negative, then stop.

$$FB = 1.0$$

$$C1 = \frac{MU\ (MU + 3)\left(1 + \sqrt{\frac{MU}{MU + 2}}\right)}{8 * (1 + MU)}$$

$$C2 = \sqrt{C1}$$

$$A = \left(2 * C2 * \frac{FR}{FR1}\right)^2$$

$$B = \left(\frac{FR}{FR1}\right)^2 - FB^2$$

$$C = \left(\frac{FR}{FR1}\right)^2 - 1.$$

$$D = MU * \left(\frac{FR}{FR1}\right)^2$$

$$E = MU * FB^2 * \left(\frac{FR}{FR1}\right)^2$$

$$W/WST = \sqrt{\frac{A + B^2}{A * (C + D)^2 + (E - (C * B))^2}}$$

NAMELIST INPUT FOR BEST TUNED ABSORBER PROGRAM:

```
$INVAL EI=20000000.,L=20.,MB=90.,MF=360.,W=.10,ACC=.001,
NP=100,DELTA=.01,FINIT=.1$
$INVAL L=30.$
$INVAL L=35.$
$INVAL L=40.$
$INVAL L=50.$
$INVAL L=55.$
$INVAL L=56.$
$INVAL L=57.$
$INVAL L=58.$
$INVAL L=59.$
$INVAL L=60.$
$INVAL L=61.$
$INVAL L=62.$
$INVAL L=63.$
```

NAMELIST INPUT FOR FREQUENCY MATCHED ABSORBER PROGRAM:

```
$INVAL EI=20000000.,L=20.,MB=90.,MF=360.,W=.10,ACC=.001,
NP=100,DELTA=.01,FINIT=.1$
$INVAL L=25.0$
$INVAL L=30.0$
$INVAL L=32.$
$INVAL L=34.0$
$INVAL L=35.0$
$INVAL L=36.0$
$INVAL L=37.0$
$INVAL L=38.0$
$INVAL L=38.5$
$INVAL L=39.0$
$INVAL L=39.5$
$INVAL L=40.0$
$INVAL L=40.1$
$INVAL L=40.2$
$INVAL L=40.3$
```

```
      PROGRAM  BTA(INPUT,OUTPUT,TAPE5,TAPE6,TAPE3)
C
C **********       BEST TUNED ABSORBER PROGRAM       **********
C
C       NORMALIZED DISPLACEMENT RESPONSE IS PLOTTED
C         AGAINST NORMALIZED EXCITATION FREQUENCY
C
C       READS NAMELIST INPUT FROM TAPE5
C
C       WRITES MU AND L VALUES TO TAPE3 FOR FUTURE PLOTTING
C
C       WRITES OUTPUT TO TAPE6
C
C
      COMMON/PARAMS/ EI,L,MB,MF,W,ACC,DELTA
     1,M1,K1,WST,C2,FR1,FR2,M2,F1,F2,MU
C
C       JTB COMMON IS FOR NASA-LANGLEY INTERFACE TO PLOT 10
C
      COMMON/JTB/NFR,JREQ,IBAUD,HDR,IJO,TFAC,IJTB(4)
      DIMENSION FROFR1(200),WOWST(200)
      REAL K1,L,M1,MF,MB,MU,M2
      NAMELIST/INVAL/EI,L,MB,MF,W,ACC,NP,DELTA,FINIT
      DATA PI/3.14592654/
      JREQ=2
 5    READ(5,INVAL)
      IF(EOF(5)) 400,10
 10   WRITE(6,INVAL)
      M1=MF+1.0/3.0*MB
      K1=3.0*EI/L**3.0
      WST=(M1*L*ACC)/K1
      MU=2.0/((W/WST)**2-1.)
C WRITE MU,L TO TAPE3
      WRITE(3,500) MU,L
      M2=MU*M1
      RKOM=K1/M1
      FR1=SQRT(RKOM)
      F1=FR1/(2.0*PI)
      FB=1.0/(1.0+MU)
      FR2=FB*FR1
      F2=FB*F1
      C1=(3.0*MU)/(8.*(1.0+MU)**3)
      C2=SQRT(C1)
      WRITE(6,530)
      WRITE(6,500)M1,K1,WST,MU,M2
      WRITE(6,540)
      WRITE(6,500) C2,FR1,FR2,F1,F2
      WRITE(6,510)
      DO 100  I= 1,NP
      FROFR1(I)=FINIT + DELTA*(I-1)
      GSQ=FROFR1(I)**2.0
      A=(2.0*C2*FROFR1(I))**2.0
      B=(GSQ) - FB**2.0
      C=GSQ-1.0
      D=MU*GSQ
      E=MU*FB**2.0*GSQ
      WOWST(I)=SQRT((A+B**2)/(A*(C+D)**2.0+(E-(C*B))**2.0))
      WRITE(6,500) FROFR1(I),WOWST(I)
100   CONTINUE
      WMAX=0.
C FIND MAXIMUM VALUES
      DO 200 I= 1,NP
      IF(WOWST(I).LT.WMAX) GO TO 200
      WMAX=WOWST(I)
```

```
      FMAX=FROFR1(I)
 200  CONTINUE
      WRITE(6,550) FMAX,WMAX
C    CALL PLOTIT TO PLOT VALUES
      CALL PLOTIT(FROFR1,WOWST,NP,FMAX,WMAX)
C    CALL TINPUT TO PAUSE BETWEEN FRAMES
      CALL TINPUT(I)
 300  CONTINUE
      GO TO 5
 400  CONTINUE
      CALL FINITT(0,0)
      STOP
 500  FORMAT(5E15.5)
 510  FORMAT(6X,6HFR/FR1,10X,6HW/WST ,14X,2HM2)
 530  FORMAT(8X,2HM1,14X,2HK1,14X,3HWST,14X,2HMU,14X,2HM2)
 540  FORMAT(8X,2HC2,14X,3HFR1,13X,3HFR2,12X,2HF1,14X,2HF2)
 550  FORMAT(2X, 10HMAXIMUM = ,F15.5)
      END
```

```
      PROGRAM  FMA(INPUT,OUTPUT,TAPE5,TAPE6,TAPE3)
C
C   **********        FREQUENCY MATCHED ABSORBER      **********
C
C      READS NAMELIST INPUT FROM TAPE5
C      WRITES MU AND L VALUES TO TAPE3 FOR FUTURE PLOTTING
C      WRITES OUTPUT TO TAPE6
C
C        JTB COMMON IS FOR NASA-LANGLEY INTERFACE TO PLOT10
C
      COMMON/PARAMS/ EI,L,MB,MF,W,ACC,DELTA
     1,M1,K1,WST,C2,FR1,FR2,M2,F1,F2,MU
      COMMON/JTB/NFR,JREQ,IBAUD,HDR,IJO,TFAC,IJTB(4)
       DIMENSION FROFR1(200),WOWST(200)
       REAL K1,L,M1,MF,MB,MU,M2,MU1,MU2
       NAMELIST/INVAL/EI,L,MB,MF,W,ACC,NP,DELTA,FINIT
       DATA PI/3.14592654/
       JREQ=1
 5     READ(5,INVAL)
       IF(EOF(5)) 400,10
 10    WRITE(6,INVAL)
       M1=MF+1.0/3.0*MB
       K1=3.0*EI/L**3.0
       WST=(M1*L*ACC)/K1
       X=W/WST
       TEMP1=X**2.0-4.0*X-1.0
       TEMP1S=TEMP1**2.0
       TEMP2=SQRT(TEMP1S-16.0*X)
       MU1=(TEMP1+TEMP2)/(4.0*X)
       MU2=(TEMP1-TEMP2)/(4.0*X)
       IF ((MU1 .LT. 0) .AND. (MU2 .LT.0)) GO TO 450
       IF(MU1.LE.0) MU=MU2
       IF(MU2.LE.0) MU=MU1
       IF((MU1.GT.0) .AND.( MU1.LT.MU2)) MU=MU1
       IF((MU2.GT.0) .AND.( MU2.LT.MU1))MU=MU2
C    WRITE VALUES OF MU AND L TO TAPE3
       WRITE(3,500) MU,L
       M2=MU*M1
       RKOM=K1/M1
       FR1=SQRT(RKOM)
       F1=FR1/(2.0*PI)
       FB=1.0
       FR2=FB*FR1
       F2=FB*F1
       C1=MU*(MU+3.0)*(1.0+SQRT(MU/(MU+2.)))/(8.*(1.+MU))
       C2=SQRT(C1)
       WRITE(6,530)
       WRITE(6,500)M1,K1,WST,MU,M2
       WRITE(6,540)
       WRITE(6,500) C2,FR1,FR2,F1,F2
       WRITE(6,510)
       DO 100  I= 1,NP
       FROFR1(I)=FINIT + DELTA*(I-1)
       GSQ=FROFR1(I)**2.0
       A=(2.0*C2*FROFR1(I))**2.0
 18    B=(GSQ) - FB**2.0
       C=GSQ-1.0
       D=MU*GSQ
       E=MU*FB**2.0*GSQ
```

```
      WOWST(I)=SQRT((A+B**2)/(A*(C+D)**2.0+(E-(C*B))**2.0))
      WRITE(6,500) FROFR1(I),WOWST(I)
100   CONTINUE
      RMAX=0.
      DO 200 I= 1,NP
      IF(WOWST(I).LT.RMAX) GO TO 200
      RMAX=WOWST(I)
      FMAX=FROFR1(I)
200   CONTINUE
      WRITE(6,550) FMAX,RMAX
      CALL PLOTIT(FROFR1,WOWST,NP,FMAX,RMAX)
      CALL TINPUT(I)
300   CONTINUE
      GO TO 5
400   CONTINUE
      CALL FINITT(0,0)
      STOP
450   WRITE(6,451) MU
451   FORMAT(20H MU IS NEGATIVE      ,E20.5)
      GO TO 300
500   FORMAT(5E15.5)
510   FORMAT(6X,6HFR/FR1,10X,6HW/WST )
530   FORMAT(8X,2HM1,14X,2HK1,14X,3HWST,14X,2HMU,14X,2HM2)
540   FORMAT(8X,2HC2,14X,3HFR1,13X,3HFR2,12X,2HF1,14X,2HF2)
550   FORMAT(2X, 10HMAXIMUM = ,F15.5)
      END
```



Figure 1 - Fifty-five meter diameter offset wrap rib space antenna concept.



a) Beam and Feed Mass Configuration



b) Free Body Diagram



c) Equivalent Two Degree of Freedom System

Figure 2 - Simplified Analysis Model.

Theorem: For a 2 DOF System, Two Points Exist P and Q Whose Abscissae and Ordinates are Independent of Damping



Figure 3 - Schematic illustrating fixed point theorem (taken from ref. 2).



Figure 4 - General Design Curves for Best Tuned and Frequency Matched Absorbers.

Required Tip Deflection Tolerance: $w = \pm 10$ cm

Maximum Orbiter Acceleration: $\ddot{\Theta} = 0.001$ rad/sec^2

General Dynamics Four-Longeron Space Mast



MAST PROPERTIES

EI	m	ℓ
2×10^7 N.m^2	90 kg	10 - 100 m

FEED PROPERTIES

$M = 360$ kg

Figure 5 - Space Mast Experiment Geometry and Requirements.



Figure 6. Vibration absorber mass required for best-tuned space mast absorber.



Figure 7. Vibration absorber mass required for frequency-matched space mast absorber.



a) L = 58 M

Figure 8. Behavior of optimized absorber for large absorber masses.



b) L = 59 M

Figure 8. Continued.



c) L = 60 M

Figure 8. Concluded.



Figure 9 - Static force required by in-board actuator to produce constant tip deflection.

1. Report No. NASA TM 84488	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle PRELIMINARY SIZING OF VIBRATION ABSORBER FOR SPACE MAST STRUCTURES		5. Report Date May 1982
		6. Performing Organization Code
7. Author(s) Michael F. Card, Harvey G. McComb, Jr., and Susan W. Peebles		8. Performing Organization Report No.
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665		10. Work Unit No. 506-53-43-04
		11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		13. Type of Report and Period Covered Technical Memorandum
		14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

A simple method of sizing a vibration absorber for a large, cantilevered flexible mast is presented. The method is based on Den Hartog's vibration absorber theory for two-degree-of-freedom systems. Generalized design curves are presented as well as specific numerical results for a candidate space experiment in which a long flexible antenna mast is attached to the shuttle orbiter and dynamically excited by orbiter accelerations. Results indicate that for large flexible masts, the mass of the vibration absorber required to meet stringent tip deflection tolerances becomes prohibitively large.

17. Key Words (Suggested by Author(s)) Large space structures Space masts Space flight experiments Vibration absorption	18. Distribution Statement Unclassified - Unlimited Subject Category 18

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	31	A03

N-305